

Perl Short Course: Some Extra Notes

Duncan C. White (d.white@imperial.ac.uk)

Dept of Computing,
Imperial College London

January 2015

- Here's some material that got removed from the Perl course, or never quite made it - might be interesting:
 - **Fifth Lecture: DBI films example, handling null/undef better**
 - **Sixth Lecture: Person and Programmer version 4**
 - **Moose: a new OO system for Perl**
 - **How CSG use Perl**

- In the DBI films example eg6, slide 19, we found a problem: one film record had a null **length** field (which became **undef** in Perl). This caused a warning, which we fixed in the lecture by:

```
while( my $record = $sth->fetchrow_hashref )
{
    print "Title:    $record->{title}\n";
    print "Director: $record->{director}\n";
    print "Origin:    $record->{origin}\n";
    print "Made:      $record->{made}\n";
    my $length = $record->{length} // '';
    print "Length:   $length\n";
    print "-" x 30 . "\n";
}
$sth->finish;
```

- Perhaps *any of the fields* could have been null, could we deal with all of them better? Sounds like a job for map!

```
while( my $record = $sth->fetchrow_hashref )
{
    my( $title, $director, $origin, $made, $length ) =
        map { $record->{$_} // '' }
            qw(title director origin made length);
    print "Title:    $title\n";
    print "Director: $director\n";
    print "Origin:    $origin\n";
    print "Made:      $made\n";
    print "Length:   $length\n";
    print "-" x 30 . "\n";
}
$sth->finish;
```

- Or, if you preferred, use `map` to generate a new hash:

```
while( my $record = $sth->fetchrow_hashref )
{
    my %f = map { $_ => $record->{$_} // '' } keys %$record;
    print "Title:   ${f{title}}\n";
    print "Director: ${f{director}}\n";
    print "Origin:   ${f{origin}}\n";
    print "Made:     ${f{made}}\n";
    print "Length:  ${f{length}}\n";
    print "-" x 30 . "\n";
}
$sth->finish;
```

- Or we could use a procedural map to modify `$_$record`:

```
while( my $record = $sth->fetchrow_hashref )
{
    map { $record->{$_} //= '' } keys %$record;
    print "Title:   $title\n";
    print "Director: $director\n";
    print "Origin:   $origin\n";
    print "Made:     $made\n";
    print "Length:  $length\n";
    print "-" x 30 . "\n";
}
$sth->finish;
```

- Whichever way we decide to do it, we might want our general `sql_foreach()` function to do such de-nulling once and for all.

- In the sixth lecture, we presented a lengthy class example about people (class **Person**) with names, sexes and ages, and programmers (subclass **Programmer**) with additional programming language skills.
- Our final version **programmer-v3** got Duncan's skills back by using *constructor chaining*.
- Isn't there a better way? Well, the only thing varying per-class appears to be the set of data fields which we want to initialize, and their default values. Remove **Programmer's** constructor, and generalise **Person's** constructor as follows:

```

fun new( $class, %arg ) {
    my $obj = bless( {}, $class );
    my %default = $obj->_defaultvalues;
    while( my($datum,$value) = each(%default) )
    {
        $obj->{$datum} = $arg{$datum} // $value;
    }
    return $obj;
}

```

- Now, each class defines a private `_defaultvalues()` method, listing the default values of all the initializable data fields:

```

method Person::_defaultvalues { return (NAME=>"Shirley", SEX=>"f", AGE=>26); }

```

- Continuing:

```
method Programmer::_defaultvalues
{
    return ( NAME=>"Shirley", SEX=>"f", AGE=>26, SKILLS=>{java=>"ok"} );
}
```

- These methods allow a single generic `Person::new` constructor to initialize all the desired data fields. Of course, we are still repeating all the defaults in each subclass.
- Can we fix this? Yes, but back to method chaining!

```
method Programmer::_defaultvalues {
    my %default = $self->Person::_defaultvalues;
    $default{SKILLS} = { java => "ok" };
    return %default;
}
```

- More generically, we can write the chained method call as:

```
my %default = $self->SUPER::_defaultvalues;
```

to call the first available parental `_defaultvalues()` method.

- This is the final version **programmer-v4** in the Lecture 6 tarball.

As an alternative to Perl's `bless()` based objects and classes, developers have built `Moose`, a whole new OO system for Perl.

- To give you a flavour of what `Moose` can do, lecture 6's `Person` class could be written:

```
package MoosePerson;
use strict;
use warnings;
use Function::Parameters qw(:strict);
use Moose;

has 'name' => ( is => 'rw', isa => 'Str', default => 'Shirley' );
has 'sex'  => ( is => 'rw', isa => 'Str', default => 'f' );
has 'age'  => ( is => 'rw', isa => 'Int', default => 26 );

method as_string
{
    my $class = ref($self);
    my $name  = $self->name;
    my $age   = $self->age;
    my $sex   = $self->sex;
    return "$class( name=$name, age=$age, sex=$sex )";
}

# stringification
use overload '""' => \&overload_as_string;
fun overload_as_string( $list, $x, $y ) # don't care about last 2 params
{
    return $list->as_string;
}

1;
```

- The example program (**eg2**) is virtually unchanged (only change: the new parameter keys are now lower-case):

```
use MoosePerson;
my $dunc = MoosePerson->new( name => "Duncan", age => 45, sex => "m" );
print "$dunc\n";
$dunc->age( 20 ); $dunc->name( "Young dunc" );
print "$dunc\n";
```

- Let's see a Moose version of Programmer:

```
package MooseProgrammer;
use strict;
use warnings;
use Function::Parameters qw(:strict);
use Moose;

extends 'MoosePerson';
has 'skills' => (
    is      => 'rw',
    isa     => 'HashRef',
    default => sub { return { java => 'ok' } },
);
method skills_as_string {          # additional method
    my $sk = $self->skills;
    my @str = map { sprintf( "%s:%s", $_, $sk->{$_} ) } sort(keys(%$sk));
    return "{" . join(", ", @str) . "}";
}
around 'as_string' => fun( $orig, $self ) {
    my $pers = $self->$orig(@_); $pers =~ s/ \\\$//;
    my $skills = $self->skills_as_string;
    return "$pers, skills=$skills)";
};
1;
```


- Inheritance is done through the `extends 'ParentClass'` syntax. Moose also offers nice method wrapping - `around` is one such feature.
- Let's see a Moose version of **eg3a**:

```
use MooseProgrammer;
my $dunc = MooseProgrammer->new( name => "Duncan", age => 45, sex => "m",
                                skills => {
                                    "C" => "godlike",
                                    "perl" => "godlike",
                                    "C++" => "ok",
                                    "java" => "minimal"
                                } );

print "$dunc\n";
$dunc->age( 20 );
$dunc->name( "Young dunc" );
$dunc->skills( { "C" => "good", "prolog" => "good" } );
print "$dunc\n";
```

- As expected, when run, that says:

```
MooseProgrammer( name=Duncan, age=45, sex=m, skills={C:godlike, C++:ok, java:minimal, perl:godlike} )
MooseProgrammer( name=Young dunc, age=20, sex=m, skills={C:good, prolog:good} )
```

- Moose also supports user-defined types, roles, delegation etc. It does seem genuinely easier to use than blessed reference OO, and does a lot of the boilerplate dull stuff for you.

CSG use Perl a great deal:

- When we unpack a new machine, we enter inventory, hostname, IP address and MAC address details into a Postgres database - via a **web database interface (a Perl CGI script)**. We also add configuration information about the new desktop in another database table (host classes, what type of machine it is) and do a small amount of extra configuration.
- When we plug the new machine into the network and turn it on, we boot from a Linux USB key which NFS mounts a “root filesystem” and then starts running a **CSG-custom installation and maintenance system, entirely written in Perl**.
 - First, this chooses which disk(s) to use as the boot disk.
 - Then it partitions those disks using whichever partitioning scheme the machine’s host class info indicates should be used (the partitioning scheme is written in Perl too), optionally creates Linux logical volumes on the partitions, creates filesystems on those partitions/logical volumes, and mounts them.
 - Then it replicates the NFS root filesystem to the fresh root filesystem, and switches into it.

- Continuing:
 - Then successive maintenance scripts copy in numerous configuration files (universal to DoC, or specific to the machine's host classes), and install hundreds (servers) or thousands (desktops) of packages - selecting the appropriate package list for that type of machine (via host classes).
 - Finally, the machine boots, and on first proper boot is a full member of DoC, running the right services, having the right packages, knowing about DoC users etc etc.
 - The maintenance system regularly runs thereafter, keeping the machine up to date, installing new kernels, new packages, tweaking config files when we decide they need tweaking, etc etc.
- **Lexis, our locally developed exam lockdown system**, is entirely written in Perl - the client code that chats the Lexis custom protocol, the Lexis server (that uses Perl threads!), and a separate Perl/Tk status monitor.
- Plus a million Perl “helper” scripts, Perl one liners.
- Conclusion: **Perl code runs DoC.**